VI.5.3B OPERATIONAL FORECAST PROGRAM HYDROLOGIC COMMAND LANGUAGE
(HCL) INPUT EXAMPLES

Introduction

This Section contains sample Hydrologic Command Language (HCL) input
for a variety of typical runs of the Operational Forecast Program
(FCST) .

The examples not only illustrate the HCL input for typical runs but
also show various ways in which the FCST program can be used.

The examples are primarily intended to show runs involving the use of
more than one Function and illustrate the relationship between
Functions. Examples of the input for single Function runs are given
under each Function in Section VI.5.3C [Hyperlink].

In all of the examples it is assumed that no Local HCL defaults have
been set except that the output time zone (NOUTZ) and daylight savings
switch (NOUTDS) Techniques are assumed to be set to meet the user's
requirements. In reality the user would set default values to meet
other Local requirements and reduce the HCL input.

Examples

Example 1. Preprocessors for entire user area

This example shows the HCL input for a run of all the preprocessors
for the entire user area. Two days of observed data are included
(most recent day plus the inclusion of late observations from the
previous day), but only the most recent day is displayed. Five days
of future MAPX, MAT, MAPE and RRS time series values are produced. No
future MAP values are generated. The MAP Function uses a convective
radius of 20 miles to limit the search for point estimators and uses
MDR data to time distribute daily MAP estimates into 6 hour values.
The last date with observed data defaults to today at 12Z. The
following results are displayed: observed and estimated non-zero
daily precipitation, all 6 hour precipitation, MDR data, observed
temperature data, PE station data and all RRS observations. There are
2 Carryover Groups that require MAP computations.

@SETOPTIONS
STARTRUN *-02
ENDRUN *+05
PRLASTDY
CGROUP USERCG1
CONVEC 20.
MDRDIST
PRTPP24 (2)
PRTPP6
PRTMDR
PRTTINST
PRTT24

STNPE (2)
RRSALLOB
@COMPUTE MAP
@SETOPTIONS
CGROUP USERCG2
PRTPP24 (0)
PRTPP6 (0)
PRTMDR (0)
@COMPUTE MAP
@COMPUTE MAPX
@COMPUTE MAT
@COMPUTE MAPE
@COMPUTE RRS
@STOP

If the preprocessors were to be executed routinely in this manner then
an HCL Procedure could be created to simplify the input even further.
Since the use of the convective radius option would change depending
on storm type then the CONVEC Technique would be left out of the
Procedure. Thus if an HCL Procedure named ALLPP was created then the
same run as above could be made by entering:

@SETOPTIONS
CONVEC 20.
@EXECUTE ALLPP

Example 2. Forecast Group run

This example shows the HCL input for a run of a single Forecast Group.
First the MAP Function is run for the Forecast Group to update the MAP
time series based on precipitation reports that have been received
since MAP was last run. The MAP Function always estimates missing
station data for the entire user area, but MAP time series will only
be computed for the area specified. The other preprocessors are
always run for the entire area and could be added to this example if
needed. For the MAP Function, the options used are: the Tulsa data
estimation method, MDR used to fill in zero amounts, 6 hour stations
to determine time distribution and displays of the daily and 6 hour
data.

For the Forecast Function (FCEXEC) several runtime modifications are
entered to modify the runoff, discharges and model variables. If this
Forecast Group is downstream of other Forecast Groups, the upstream
inflows are obtained from the most recent runs of these upstream
areas.

In this example it is assumed that observed data are available through
0Z of the current day (i.e. this is an evening forecast run). MAP is
only run for the latest 12 hours, whereas FCEXEC starts 2 days ago.
The MAP output will be on unit 8, whereas FCEXEC output will be on
unit 6.

@SETOPTIONS

LSTCMPDY *00Z
STARTRUN *

FGROUP BASINX
ESTTULSA
MDREST24 (1)
PRTPP24 (2)
PRTPP6
PPPRINT (8)
@COMPUTE MAP
@SETOPTIONS
STARTRUN *-02
ENDRUN *+05
MOD
.ROCHNG 030118%
SEG1 .4 .2 0 .8
SEG2 0 .4 1.2
.TSCHNG 0229 0302
SEGUP FPX QINE 6 400 800 1000 1500 2000 1700 1500
SEGX FPY STG 12 12.4 15 18
.WECHNG 0301

SEGD 6.7
SEGM 8.1
SEGB 4.7

.UHGCHNG 0302
SEG10 0 2 7 10 7 4 2 1 / UP
SEG10 4 8 7 5 3 1 / DOWN
ENDMOD

@COMPUTE FCEXEC

@STOP

Actual, rather than relative, dates are required for MODS so that if
the MOD cards are left in the HCL input stream on subsequent runs, the
MODS will only apply to the periods for which they were intended.

A generalized HCL Procedure also could be defined for use with any
Forecast Group through the use of symbolic parameters in the Procedure
definition. The default values of the symbolic parameters are
assigned when the Procedure is defined. These values then can be
changed at run time. An example of the definition of such a Procedure
is as follows:

@DEFINEL PROCEDURE FGRUN
PXEST=CONVEC
MDRUSE=N
MDRLMT=0
FGROUP=DUMMY
STARTMAP=*-01
STARTFC=*-02
LASTOB=*
@SETOPTIONS
LSTCMPDY &LASTOB
STARTRUN &STARTMAP
FGROUP &FGROUP
&PXEST
MDREST24 (&MDRUSE) &MDRLMT
PRTPP24 (2)
PRTPP6

PPPRINT (8)
@COMPUTE MAP
@SETOPTIONS

STARTRUN &STARTEC

ENDRUN *+05
@COMPUTE FCEXEC
@STOP

END

The partial area run for Forecast Group BASINX shown at the beginning
of this example can be executed using the generalized FGRUN Procedure
with the following input:

@SETOPTIONS

MOD

All MODS in effect

ENDMOD

QEXECUTE FGRUN LASTOB=*007Z STARTMAP=* FGROUP=BASINX
PXEXT=ESTTULSA MDRUSE=Y STARTFC=*-01

Example 3. Save carryover run

Carryover can be saved only when an FCEXEC Carryover Group run is
made. Normally a carryover save run is made after all the data are
posted and processed and all adjustments (MODS) are determined
(sometimes referred to as the 'final' run of the day). A carryover
save run will take longer to complete than a CGROUP run where
carryover is not saved due to the extra computations and disk accesses
required.

In the example, the run starts 2 days ago so that any late data from
the previous day is included in the carryover values. Carryover is
saved for today, but carryover for yesterday, if available, will be
automatically updated. No future data are included since these data
do not affect today's carryover. The MODS that are currently in
effect are stored by Forecast Group names in files. All FCEXEC
printout (hydrograph displays, etc.) is turned off. 1In addition to
the FCEXEC Function, two other Functions are included in this example.
The CGSTATUS Function displays a table showing which dates carryover
is available and the PRINTOPS Function displays a summary of current
conditions for the snow and Sacramento models.

@SETOPTIONS
CGROUP USERGG
STARTRUN *-02
ENDRUN *

SAVETDY

PRINTOUT (0)

MOD
.INCLUDE BASINA
.INCLUDE BASINB
.INCLUDE BASINC
.INCLUDE BASIND
ENDMOD

@COMPUTE FCEXEC

@COMPUTE CGSTATUS
@SETOPTIONS

PRNTCGRP USERCG

PRNTLAST 1

PRNTOPER SNOW-17 SAC-SMA
@COMPUTE PRINTOPS
@STOP

An HCL Procedure could be created from these instructions, named FINAL
for example, to reduce the HCL input. The input for the carryover

save run would then become:

@EXECUTE FINAL

Example 4 - QPF or contingency run

No future precipitation data were used in the previous examples. This
example shows a run where future MAP data are input and translated
into time series by the FMAP Function. The future MAP time series are
then used by the FCEXEC Function to generate QPF or contingency
forecasts at points of interest.

The HCL input could be:

@SETOPTIONS

LSTCMPDY *

MOD
.FMAP6 030318%Z
AREA17-AREA8 .5 2.0 1.0 0.5
AREA6 01. 1.5 1. .5
AREA29 O .2 .5 2. 1. .5
ENDMOD

@EXECUTE QPFFPX

where the HCL Procedure contains the instructions to produce a run
using future precipitation for the river network above a certain flow
point. The Procedure QPFFPX contains:

@SETOPTIONS
FGROUP FPX
INCLUDE QPFOPT
PLOTHYD (FPX)
PLOTHYD (FPM)
PLOTHYD (FPZ)

@COMPUTE FMAP

@COMPUTE FCEXEC

@SETOPTIONS
FUTPRECP (0)
PRINTOUT (0)

@COMPUTE FCEXEC

@STOP

The named option QPFOPT contains the QPF related options for FCEXEC
and the ENDRUN value for both FMAP and FCEXEC.

STARTRUN *-01
ENDRUN *+10
FUTPRECP (1)
PLOTHYD (0)

In this example future MAP time series values are first input via the
.FMAP6 MOD to define the QPF pattern. The special Forecast Group FPX
is then run. This special Forecast Group contains the Segments needed
to produce a QPF forecast at point X, a point where QPF forecasts are
frequently requested. The QPFOPT named option sets the run period,
turns the future precipitation flag 'on' for the FCEXEC Function and
turns all hydrograph plots 'off'. Next hydrograph plots are turned
'on' for only those points where a QPF forecast is needed. After the
QPF forecast is computed, the FCEXEC Function is run again with future
precipitation and printout 'off' in order to reset all time series in
the Processed Data Base back to their pre-QPF condition. If this last
step is not included then a subsequent regular forecast run for a
downstream FGROUP could use upstream inflows stored in the Processed
Data Base that contain QPF generated flows.

